
PORTA27 
Display 

  

(HP-41CX, Hewlett Packard 1983 and DM41X, SwissMicros 2020) 

 

Overview1 

Program PORTA27 implements an encryption variation by Giambattista della Porta, known 
as polyalphabetic cipher, so called bigraphic substitution. Porta published De Furtivis Literarum Notis in 
1563. This program is written for the HP-41CX as minimisation of the program in chapter 7 of the 
book: Kryptologie (HP-41 C/CV): Chiffrierung, Textverschlüsselung, Datenschutz, Morsealphabet; Alt, 
Helmut; Schumny, Harald [Hrsg.], Braunschweig [u.a.]. Vieweg. 1983. Vieweg-Programm-bibliothek 
Taschenrechner Band 7. ISBN 3-528-04256-7. 

 
 

Formula 

Coding according to Porta is done by looking up the value of two consecutive characters in a table (see 
Encryption Table). For example the word “PO” can be coded as 420. If a word has an odd number of 
characters, a space is appended for completion of the Porta coding. The table lookup can be computed 
by following below algorithms: 

 

FORWARD 

Taking the two consecutive characters in a word C1 and C2, the Porta code cp can be 
determined as follows: 

cp = 27.(C1 – 1) + C2 C1, C2 are A..Z,space 

cp = 27.C1 + C2 – 1819 
Substituting Ci = Ci + (65-1),  

where 65 is char value of A 

BACKWARD 

Taking the Porta code cp, the two consecutive characters in a word C1 and C2 can be 
determined as follows: 

C1 = INT(0,99 + cp/27) 
C2 = INT(0,99 + 27.FRC(cp/27)) 

cp is 1..729 

 

The lookup table shows a base of 27. More bases are possible if more characters need to be encrypted. 

An example of coding the word HOOXIES is shown in the encryption table which translates the word into 

a series of cp values: 204, 402, 221 and 513. 
 

 
1 This program is copyright and is supplied without representation or warranty of any kind. The author assumes no responsibility and shall have no 
liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof 

https://www.swissmicros.com/
http://en.wikipedia.org/wiki/Giambattista_della_Porta
http://en.wikipedia.org/wiki/Polyalphabetic_substitution
http://www-history.mcs.st-andrews.ac.uk/Biographies/Porta.html
http://www.amazon.de/gp/product/3528042567?ie=UTF&linkCode=as2&camp=1638&creative=6742&tag=liball-21&creativeASIN=3528042567
http://www.amazon.de/gp/product/3528042567?ie=UTF&linkCode=as2&camp=1638&creative=6742&tag=liball-21&creativeASIN=3528042567
http://www.amazon.de/gp/product/3528042567?ie=UTF&linkCode=as2&camp=1638&creative=6742&tag=liball-21&creativeASIN=3528042567

Encryption Table 

Cross first character in the left column with the second character in the top row. The character sequence 
is the encrypted code. For example, the word HOOXIES can be coded as follows: 

 
1. "HO", in table is: 204 

2. "OX", in table is: 402 

3. "IE", in table is: 221 

4. "S ", in table is: 513 

Example (1) 

Please note my default FIX 5 setting in de examples below for the codes.  

KEYSTROKES DISPLAY COMMENTS 

  Run forward program 

[XEQ] [ALPHA]P27F[ALPHA] 
Enter Ci values,  
e.g. PORTA CODING ON HOOXIES 

PORTA CODING ON HOOXIES[R/S]  First value cp 

[R/S]  Second value cp 

[R/S]  Third value cp 

[R/S]  Fourth value cp 

[R/S]  Fifth value cp 

[R/S] 

 
 Sixth value cp 

[R/S]  Seventh value cp 

[R/S]  Eighth value cp 

[R/S]  Nineth value cp 

[R/S]  Tenth value cp 

[R/S]  Eleventh value cp 

[R/S]  Twelfth value cp 

[R/S]  
Completed, code is:  
420-479-27-69-90-358-513-392-710-393-630-127 

[R/S]  Enter Ci values, e.g. HP 

HP[R/S]  Value cp 

[R/S]  Completed, code is: 205 

[R/S]  Enter other Ci values 

 

 

Example (2) 

Below example decrypts a numerical sequence back to a text. 
 

KEYSTROKES DISPLAY COMMENTS 

  Run backward program 

[XEQ] [ALPHA]P27B[ALPHA]  Enter first cp value, e.g. 204 

204[R/S]  Enter second cp value, e.g. 402 

402[R/S]  Enter third cp value, e.g. 221 

221[R/S]  Enter fourth cp value, e.g. 513 

513[R/S]  Stop new values, press R/S to continue 

[R/S]  Completed indicated with dot at the end 

[R/S]  Start with new cp value(s) 

Program Listing 

P27F makes use of an optimised (forward) algorithm that includes the character values of alphas A..Z 
(65..90). However, the space value (32) does not fit so nicely in this sequence, the formulae is non-linear. 
In LBL 02 this is intercepted by checking against 32. If the 
ATOX value exceeds 32, it can proceed according to the forward algorithm, otherwise it substitutes the 
value 91 representing the character value of a space in the optimised (forward) algorithm. 
P27B (backward) has to cope with the same non-linear problem of the space. In LBL 03 it checks 
whether the calculated character value is zero. This happens for character 27 (space) and results in cp/27 
getting very close to zero. The program checks whether X=0? and appends a space to the text string, 
otherwise it converts XTOA (atter first adding 64 to any of the possible values 1..26). 

 01▪LBL "P27F" 

 02▪LBL 03 

 03 "TXT=?" 

 04 AON 

 05 PROMPT 
 06 AOFF 

 07 ALENG 

 08 STO 00 

 09 2 

 10 MOD 

 11 ST+ 00 

 12 LASTX 

 13 ST/ 00 

 14▪LBL 00 

 15 XEQ 02 

 16 27 

 17 * 
 18 STO 02 

 19 XEQ 02 

 20 RCL 02 

 21 + 

 22 1819 

 23 - 

 24 STOP 
 25 DSE 00 

 26 GTO 00 

 27 "OK" 

 28 PROMPT 

 29 GTO 03 

 30▪LBL 02 

 31 32 

 32 ATOX 

 33 X>Y? 

 34 RTN 

 35 91 

 36 END  
 

(69 bytes) 

 01▪LBL "P27B" 

 02 "CODE=?" 

 03 AVIEW 

 04 CLA 

 05▪LBL 00 
 06 CF 22 

 07 STOP 

 08 FC? 22 

 09 GTO 02 

 10 27 

 11 / 

 12 STO 01 

 13 XEQ 03 

 14 RCL 01 

 15 FRC 

 16 27 

 17 * 
 18 XEQ 03 

 19 AVIEW 

 20 GTO 00 

 21▪LBL 03 

 22 .99 

 23 + 

 24 INT 
 25 X=0? 

 26 >" " 

 27 X=0? 

 28 RTN 

 29 64 

 30 + 

 31 XTOA 

 32 RTN 

 33▪LBL 02 

 34 >"." 

 35 AVIEW 

 36 END  
 
(69 bytes)

 

 
Registers, Labels and Flags 

REGISTERS COMMENTS  LABELS COMMENTS 

R00 
Number of required lookups: 
length of the text divided by 2 

 
LBL02 Present outcome 

R01 Cp divided by 27 
 

LBL02 
Loop around the given input 
text characters 

  

FLAGS COMMENTS 

22 Check for keyboard input 

 

 

Downloads 

The RAW/TXT format of the program is available via the website: PORTA27 (in zip file). 

https://hooxies.nl/?page_id=42

